

## G03ZAF – NAG Fortran Library Routine Document

**Note.** Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

### 1 Purpose

G03ZAF produces standardized values (*z*-scores) for a data matrix.

### 2 Specification

```
SUBROUTINE G03ZAF(N, M, X, LDX, NVAR, ISX, S, E, Z, LDZ, IFAIL)
INTEGER          N, M, LDX, NVAR, ISX(M), LDZ, IFAIL
real            X(LDX,M), S(M), E(M), Z(LDZ,NVAR)
```

### 3 Description

For a data matrix,  $X$ , consisting of  $n$  observations on  $p$  variables, with elements  $x_{ij}$ , G03ZAF computes a matrix,  $Z$ , with elements  $z_{ij}$  such that:

$$z_{ij} = \frac{x_{ij} - \mu_j}{\sigma_j}, \quad i = 1, 2, \dots, n; \quad j = 1, 2, \dots, p,$$

where  $\mu_j$  is a location shift and  $\sigma_j$  is a scaling factor. Typically  $\mu_j$  will be the mean and  $\sigma_j$  will be the standard deviation of the  $j$ th variable and therefore the elements in column  $j$  of  $Z$  will have zero mean and unit variance.

### 4 References

None.

### 5 Parameters

- 1: N — INTEGER *Input*  
*On entry:* the number of observations in the data matrix,  $n$ .  
*Constraint:*  $N \geq 1$ .
- 2: M — INTEGER *Input*  
*On entry:* the number of variables in the data array X.  
*Constraint:*  $M \geq \text{NVAR}$ .
- 3: X(LDX,M) — *real* array *Input*  
*On entry:* X( $i, j$ ) must contain the  $i$ th sample point for the  $j$ th variable,  $x_{ij}$ , for  $i = 1, 2, \dots, n$ ;  $j = 1, 2, \dots, M$ .
- 4: LDX — INTEGER *Input*  
*On entry:* the first dimension of the array X as declared in the (sub)program from which G03ZAF is called.  
*Constraint:*  $\text{LDX} \geq N$ .
- 5: NVAR — INTEGER *Input*  
*On entry:* the number of variables to be standardized,  $p$ .  
*Constraint:*  $\text{NVAR} \geq 1$ .

- 6:** ISX(M) — INTEGER array *Input*  
*On entry:* ISX( $j$ ) indicates whether or not the observations on the  $j$ th variable are included in the matrix of standardized values.  
 If ISX( $j$ )  $\neq$  0, then the observations from the  $j$ th variable are included.  
 If ISX( $j$ ) = 0, then the observations from the  $j$ th variable are not included.  
*Constraint:* ISX( $j$ )  $\neq$  0 for NVAR values of  $j$ .
- 7:** S(M) — *real* array *Input*  
*On entry:* if ISX( $j$ )  $\neq$  0, then S( $j$ ) must contain the scaling (standard deviation),  $\sigma_j$ , for the  $j$ th variable.  
 If ISX( $j$ ) = 0, then S( $j$ ) is not referenced.  
*Constraint:* if ISX( $j$ )  $\neq$  0, then S( $j$ )  $>$  0.0 for  $j = 1, 2, \dots, M$ .
- 8:** E(M) — *real* array *Input*  
*On entry:* if ISX( $j$ )  $\neq$  0, then E( $j$ ) must contain the location shift (mean),  $\mu_j$ , for the  $j$ th variable.  
 If ISX( $j$ ) = 0, then E( $j$ ) is not referenced.
- 9:** Z(LDZ,NVAR) — *real* array *Output*  
*On exit:* the matrix of standardized values ( $z$ -scores),  $Z$ .
- 10:** LDZ — INTEGER *Input*  
*On entry:* the first dimension of the array Z as declared in the (sub)program from which G03ZAF is called.  
*Constraint:* LDZ  $\geq$  N.
- 11:** IFAIL — INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.  
*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors detected by the routine:

IFAIL = 1

- On entry, N  $<$  1,
- or NVAR  $<$  1,
- or M  $<$  NVAR,
- or LDX  $<$  N,
- or LDZ  $<$  N.

IFAIL = 2

- On entry, there are not precisely NVAR elements of ISX  $\neq$  0.

IFAIL = 3

- On entry, ISX( $j$ )  $\neq$  0 and S( $j$ )  $\leq$  0.0 for some  $j$ .

## 7 Accuracy

Standard accuracy is achieved.

## 8 Further Comments

Means and standard deviations may be obtained using G01AAF or G02BXF.

## 9 Example

A 4 by 3 data matrix is input along with location and scaling values. The first and third columns are scaled and the results printed.

### 9.1 Program Text

**Note.** The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      G03ZAF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5,NOUT=6)
      INTEGER          NMAX, MMAX
      PARAMETER       (NMAX=4,MMAX=3)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, J, M, N, NVAR
*      .. Local Arrays ..
      real            E(MMAX), S(MMAX), X(NMAX,MMAX), Z(NMAX,MMAX)
      INTEGER          ISX(MMAX)
*      .. External Subroutines ..
      EXTERNAL        G03ZAF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G03ZAF Example Program Results'
*      Skip headings in data file
      READ (NIN,*)
      READ (NIN,*) N, M, NVAR
      IF (M.LE.MMAX .AND. N.LE.NMAX) THEN
        DO 20 I = 1, N
          READ (NIN,*) (X(I,J),J=1,M)
20      CONTINUE
          READ (NIN,*) (ISX(J),J=1,M)
          READ (NIN,*) (E(J),J=1,M)
          READ (NIN,*) (S(J),J=1,M)
          IFAIL = 0
*
          CALL G03ZAF(N,M,X,NMAX,NVAR,ISX,S,E,Z,NMAX,IFAIL)
*
          WRITE (NOUT,*)
          WRITE (NOUT,*) ' Standardized Values'
          DO 40 I = 1, N
            WRITE (NOUT,99999) (Z(I,J),J=1,NVAR)
40      CONTINUE
          END IF
          STOP
*
99999 FORMAT (1X,9F8.3)

```

END

## 9.2 Program Data

G03ZAF Example Program Data

```
4 3 2
15.0 0.0 1500.0
12.0 1.0 1000.0
18.0 2.0 1200.0
14.0 3.0 500.0
1      0      1
14.75 0.0 1050.0
2.50 0.0 420.3
```

## 9.3 Program Results

G03ZAF Example Program Results

Standardized Values

```
0.100  1.071
-1.100 -0.119
1.300  0.357
-0.300 -1.309
```

---